
A physiologically based toxicokinetic model for inhaled ethylene and ethylene oxide in mouse, rat, and human

Johannes Georg Filser  and Dominik Klein



Model code for uninterrupted exposures of mice, rats or humans to ET and/or EO 

using the program Berkeley Madonna X (Mach-O), version 8.3.22 for 
Mac OS X 10.6.8}

{For parameters and their abbreviations see Tables 1, 2, 3, and 10 in the article. 
A question mark stays for a number to be put in here.}

METHOD RK4
STARTTIME = 0
STOPTIME = ?
DT = 0.0001     
{DTOUT = 0.1  {Printed time steps}}

{Body weights and number of exposed animals or humans}
BW = ?     {kg, body weight of one reference animal or human (see Table 1}
BWe =  ?  {kg, average body weight of one exposed animal or human} 
Num = ?  {number of animals or humans exposed together in one chamber}

{Flows, Table 1}
Qalv=  ? /BW *BWe* Num      {L/h, 1.2 (mouse), 7.02 (rat), 300 (human)}
Qcard=  ? /BW*BWe* Num    {L/h, 1.02 (mouse), 5.04 (rat), 372 (human)}
Qr = ? * Qcard                             {L/h, 0.355 (mouse), 0.355 (rat), 0.285 (human)}
Qk = ? * Qcard                            {L/h, 0.155 (mouse), 0.155 (rat), 0.155 (human)}
Qm= ? * Qcard                            {L/h, 0.223 (mouse), 0.223 (rat), 0.250 (human)}
Qa= ? * Qcard                             {L/h, 0.017 (mouse), 0.017 (rat), 0.050 (human)}
QL= ? * Qcard                             {L/h, 0.250 (mouse), 0.250 (rat), 0.260 (human)}

{Volumes, Table 1}
Vch = ? - (BWe* Num)  {L, experimental volume of the closed-chamber with ET-exposed mice, rats, or humans}
VEOch = ? - (BWe* Num) {L, experimental volume of the closed-chamber with EO-exposed mice, rats, or humans}
Vart = ? *BWe* Num         {L, 0.0159 (mouse), 0.0241 (rat), 0.0257 (human)}
Vvnb = ? *BWe* Num       {L, 0.0331 (mouse), 0.050 (rat), 0.0533 (human)}
Vr = ? *BWe* Num            {L, 0.026 (mouse), 0.0377 (rat), 0.038 (human)}
Vk = ? *BWe* Num            {L, 0.0167 (mouse), 0.0073 (rat), 0.0044 (human)}
Vp = ? *BWe* Num            {L, 0.0073 (mouse), 0.005 (rat), 0.0076 (human)}
Vm = ? *BWe* Num           {L, 0.66 (mouse), 0.676 (rat), 0.541 (human)}
Va = ? *BWe* Num            {L, 0.10 (mouse), 0.07 (rat), 0.19 (human)}
VL = ? *BWe* Num            {L, 0.055 (mouse), 0.04 (rat), 0.026 (human)}

{Water contents, Table 1}
wfb = ?
wfr = ?
wfk = ?
wfp = ?
wfm = ?
wfa = ?
wfL = ?

{Blood-to-air and tissue-to-blood partition coeficients of ET, Table 1}
Pbair = ?
Prb = ?
Pkb = ?
Ppb = ?
Pmb = ?
Pab = ?
PLb = ?

{Species-independent blood-to-air and tissue-to-blood partition coeficients of EO, Table 1}
PEObair = 61
PEOrb = 1.03
PEOkb = 1.03
PEOpb = 1.03
PEOmb = 1.08
PEOab = 0.70
PEOLb = 0.89

{Wash-in-wash-out effect, Table 1}
fET = ?
fEO = ?

{ Species-independent rate constants of non-enzymatic pozesses, Table 2}
kEOG = 0.01248       {L/(mmol GSH*h)}
kEOh = 0.06              {1/h}

{Metabolism of ET in liver, Table 2}
k3 = ?               {1/h}
k4 = ?               {1/h}
Kmmo = ?       {mmol/L in venous liver blood}
ke = ?               {1/h}
CYPo = ?         {mmol/kg liver}

{Metabolism of EO, Table 2}
{EH in liver}
VmaxEOEH = ?     {mmol/h/kg liver}
KmapEO = ?          {mmol/L}
KmihEO = ?           {mmol/L}

{GST in liver}
VmaxgstL = ?       {mmol/h/kg liver} 
KmEOL = 9           {mmol/L, species-independent}
KmgshL = 0.1      {mmol/L species-independent}
kdgshL = 0.2       {1/h, species-independent}
CgshoL = ?           {mmol/L}

{GST in lung (p)}
Vmaxgstp =  ?       {mmol/h/kg lung}
KmEOp = 9            {mmol/L, species-independent}
Kmgshp = 0.1       {mmol/L, species-independent}
kdgshp = ?             { 1/h}
Cgshop = ?              {mmol/L}

{GST in kidneys}
Vmaxgstk = ?        {mmol/h/kg kidney}
KmEOk =  9           {mmol/L, species-independent}
Kmgshk = 0.1       {mmol/L, species-independent}
kdgshk = 0.2         {1/h, species-independent}
Cgshok =  ?            {mmol/L}

{GST in RPTG}
Vmaxgstr = ?        {mmol/h/kg RPTG}
KmEOr =  9           {mmol/L, species-independent}
Kmgshr = 0.1        {mmol/L, species-independent}
Kdgshr = 0.2         {1/h, species-independent}
Cgshor =  ?            {mmol/L}

{GST in blood}
Vmaxgstb = ?         {mmol/h/L blood}
KmEOb =  9           {mmol/L, species-independent}
Kmgshb = 0.1       {mmol/L, species-independent}
Kdgshb = 0.2        {1/h, species-independent}
Cgshob = ?             {mmol/L}

{Hb-adducts, Table 3}
khb =  ?                {L/h/g Hb, rate constant of formation of HEV; mouse, 0.8416/10000; rat, 0.46/10000; human, 0.45/10000}
ter = ?                  {h, lifespan of Hb; mouse: 960; Rat: 1440; Human: 3024}

{ DNA-Adducts} 
kDNA = 0.94/10000   {L/h/g DNA, rate constant of formation of HEG in DNA; all species}
keG =  ?                           {1/h, depurination rate constant of HEG; mouse and and rat: 0.011, human: 0.0077; Table 3}

{Initial concentration of ET; for abbreviations and dimensions see Table 10}
INIT Cair = ?                 {mmol ET/L air; the concentrations mmol/L air is obtained as the product of ppm in atmosphere with 0.0000409.}   
INIT Cp = 0
INIT Cart = 0
INIT Cvnb = 0
INIT Cr = 0
INIT Ck = 0
INIT Cm = 0
INIT Ca = 0
INIT CL = 0
INIT CYP = CYPo

{Initial concentration of EO; for abbreviations and dimensions see Table 10}
INIT CEOair =  ?           {mmol EO/L air; the concentrations mmol/L air is obtained as the product of ppm in atmosphere with 0.0000409.}  
INIT CEOp = 0
INIT CEOart = 0
INIT CEOvnb = 0
INIT CEOr = 0
INIT CEOk = 0
INIT CEOm = 0
INIT CEOa = 0
INIT CEOL = 0

{Initial GSH concentrations; For abbreviations and dimensions see Tables 2 and 10}
INIT Cgshp = Cgshop 
INIT Cgshart = Cgshob
INIT Cgshvnb = Cgshob
INIT Cgshr = Cgshor 
INIT Cgshk = Cgshok 
INIT CgshL = CgshoL

{Differential equations}
{ET}
{Air of closed exposure system}
d/dt(Cair) =(Qalv*(Cp/(Ppb*Pbair)-Cair*fET))/Vch    {mmol/L air}

{Lung}
d/dt (Cp) =(Qalv*(Cair*fET-Cp/(Ppb*Pbair))+Qcard*(Cvnb-Cp/Ppb))/Vp

{Arterial blood}
d/dt (Cart) =Qcard*(Cp/Ppb-Cart)/Vart

{Venous blood}
d/dt (Cvnb) =(Qr*Cr/Prb+Qm*Cm/Pmb+Qk*Ck/Pkb+Qa*Ca/Pab+QL*CL/PLb- Qcard*Cvnb)/Vvnb

{RPTG}
d/dt(Cr) =(Qr*(Cart-Cr/Prb))/Vr 

{Kidneys}
d/dt (Ck) =(Qk*(Cart-Ck/Pkb))/Vk

{Muscle}
d/dt(Cm) =(Qm*(Cart-Cm/Pmb))/Vm

{FAT}
d/dt(Ca) =(Qa*(Cart-Ca/Pab))/Va

{Liver}
d/dt(CYP) =ke*(CYPo-CYP)-k4*CYP*CL/(PLb*Kmmo+CL) {mmol/h/kg liver}
d/dt(CL) =(QL*(Cart-CL/PLb)-k3*CYP*VL*CL/(PLb*Kmmo+CL))/VL


{EO}
{Air of closed exposure system}
d/dt(CEOair) =(Qalv*(CEOp/(PEOpb*PEObair)-CEOair*fEO))/VEOch  {mmol/L air}

{Lung}
d/dt (CEOp) =(Qalv*(CEOair*fEO-CEOp/(PEOpb*PEObair))+Qcard*(CEOvnb-CEOp/PEOpb)-kEOh*CEOp*Vp*wfp-kEOG*Vp*Cgshp*CEOp-CEOp*Cgshp* Vmaxgstp*Vp/(Kmgshp*CEOp+KmEOp*Cgshp+CEOp*Cgshp))/Vp

{Arterial Blood}
d/dt (CEOart) =(Qcard*(CEOp/PEOpb-CEOart)-kEOh*CEOart*Vart*wfb-CEOart* Cgshart*Vmaxgstb*Vart/(Kmgshb*CEOart+KmEOb*Cgshart+CEOart*Cgshart)- kEOG*Vart*Cgshart*CEOart)/Vart

{Venous Blood}
d/dt (CEOvnb) = (Qr*CEOr/PEOrb+Qm*CEOm/PEOmb+Qa*CEOa/PEOab+QL*CEOL/PEOLb +Qk*CEOk/PEOkb-Qcard*CEOvnb-kEOh*CEOvnb*Vvnb*wfb-CEOvnb*Cgshvnb* Vmaxgstb*Vvnb/(Kmgshb*CEOvnb+KmEOb*Cgshvnb+CEOvnb*Cgshvnb)- kEOG*Vvnb*Cgshvnb*CEOvnb)/Vvnb

{RPTG}
d/dt(CEOr) =(Qr*(CEOart-CEOr/PEOrb)-kEOh*CEOr*Vr*wfr-CEOr*Cgshr*Vmaxgstr*Vr/(Kmgshr*CEOr+KmEOr*Cgshr+CEOr*Cgshr)- kEOG*Vr*Cgshr*CEOr)/Vr 

{Kidneys}
d/dt (CEOk) =(Qk*(CEOart-CEOk/PEOkb)-kEOh*CEOk*Vk*wfk-CEOk*Cgshk*Vmaxgstk*Vk/(Kmgshk*CEOk+KmEOk*Cgshk+CEOk*Cgshk)- kEOG*Vk*Cgshk*CEOk)/Vk

{Muscle}
d/dt(CEOm) =(Qm*(CEOart-CEOm/PEOmb)-kEOh*CEOm*Vm*wfm)/Vm

{Fat}
d/dt(CEOa) =(Qa*(CEOart-CEOa/PEOab)-kEOh*CEOa*Va*wfa)/Va

{Liver}
d/dt(CEOL) =(QL*(CEOart-CEOL/PEOLb)+Qih*(0.5*((CEOL-KmihEO+(k3*CYP*VL*CL)/(Qih*(CL+PLb*Kmmo))-VmaxEOEH*VL/Qih)+SQRT((CEOL-KmihEO+(k3*CYP*VL*CL)/(Qih*(CL+PLb*Kmmo))-VmaxEOEH*VL/Qih)^2+4*KmihEO*(CEOL + (k3 * CYP*VL* CL)/(Qih*(CL+PLb*Kmmo)))))-CEOL)-CEOL*CgshL*VmaxgstL*VL/(KmgshL*CEOL+KmEOL*CgshL+CEOL*CgshL)- kEOG*VL*CgshL*CEOL-kEOh*CEOL*VL*wfL)/VL

Qih =VmaxEOEH*VL/(KmapEO-KmihEO)

{GSH-metabolism}
{Lung}
d/dt(Cgshp) =(kdgshp*Vp*(Cgshop-Cgshp)-CEOp*Cgshp*Vmaxgstp*Vp/(Kmgshp*CEOp+KmEOp*Cgshp+CEOp*Cgshp)- kEOG*Vp*Cgshp*CEOp)/Vp

{Arterial blood}
d/dt(Cgshart) =(kdgshb*Vart*(Cgshob-Cgshart)-CEOart*Cgshart*Vmaxgstb*Vart/(Kmgshb*CEOart+KmEOb*Cgshart+CEOart*Cgshart)-kEOG*Vart*Cgshart*CEOart)/Vart

{Venous blood}
d/dt(Cgshvnb) =(kdgshb*Vvnb*(Cgshob-Cgshvnb)-CEOvnb*Cgshvnb*Vmaxgstb*Vvnb/(Kmgshb*CEOvnb+KmEOb*Cgshvnb+CEOvnb*Cgshvnb)-kEOG*Vvnb*Cgshvnb*CEOvnb)/Vvnb

{RPTG}
d/dt(Cgshr) =(kdgshr*Vr*(Cgshor-Cgshr)-CEOr*Cgshr*Vmaxgstr*Vr/(Kmgshr*CEOr+KmEOr*Cgshr+CEOr*Cgshr)- kEOG*Vr*Cgshr*CEOr)/Vr

{Kidneys}
d/dt(Cgshk) =(kdgshk*Vk*(Cgshok-Cgshk)- CEOk*Cgshk*Vmaxgstk*Vk/(Kmgshk*CEOk+KmEOk*Cgshk+CEOk*Cgshk)- kEOG*Vk*Cgshk*CEOk)/Vk

{Liver}
d/dt(CgshL) =(kdgshL*VL*(CgshoL-CgshL)-CEOL*CgshL*VmaxgstL*VL/(KmgshL*CEOL+KmEOL*CgshL+CEOL*CgshL)- kEOG*VL*CgshL*CEOL)/VL


{Hb-Adducts; Inits and diff.-equations}
Init CTHB = 0    {mmol HEV/g HB}
Init CHB = 0   {mmol HEV/g Hb}
d/dt(CTHB) = (Vart * CEOart  + Vvnb* CEOvnb) * khb/(Vart + Vvnb) 

d/dt(CHB) = (Vart * CEOart  + Vvnb* CEOvnb) * khb/ (Vart + Vvnb) - CTHB/ter 
{solution gives CHB at time t≤ter}

{DNA-Adducts; Inits and diff.-equations} 
Init CDNA = 0   {mmol HEG/g DNA, DNA-Adduct level}
d/dt(CDNA) = ((Vart * CEOart  + Vvnb* CEOvnb)* kDNA  - (Vart + Vvnb)*CDNA* keG) /(Vart + Vvnb)   

{Some variables, given as functions of time}
{ET concentrations}
Cairppm = Cair *24450    {ppm, ET in air}

{EO concentrations}
CEOairppm= CEOair *24450                {ppm, EO in air}
CEOvbmgL = CEOvnb * 44.05               {mg/L, EO in venous blood}
CEOvbµgL = CEOvnb * 44.05*1000    {µg/L, EO in venous blood}
CEOvbµmolL = CEOvnb *1000             {µmol/L, EO in venous blood}
CEOvbnmolL = CEOvbµmolL *1000   {nmol/L, EO in venous blood}

{Adduct levels}
nCHB = CHB * 1000000	              {nmol HEV/g Hb, HEV level }
nCDNA = CDNA * 1000000		{nmol HEG/g DNA, HEG level}
GDNA = nCDNA/0.663634451 	{pmol HEG/µmol Gua-base, HEG level}


[ ———

Jr T ——

TR ————

P —_——
Ty

B e —
i et

[ —

e e e
ey
T 1182 92060 )

i v
RS ol 15 055
RS R R
0 (e 007 015 )
e ke e

P——
[ —
e

S T 1, i i et 5.
ot )

i ) 0421 e, 0025 ).
[
e e
e e
el 2 el e
e
T




